/*******************************************************************************
03_04: Make Table X4: Effect of parent source country characteristics for second-
	generation immigrants, prioritizing father source country characteristics

Replication code for "Culture and gender allocation of tasks: source country 
	characteristics and the division of non-market work among US immigrants" in
	Review of Economics of the Household

Authors: Francine Blau, Lawrence Kahn, Matthew Comey, Amanda Eng, Pamela 
	Meyerhofer, and Alexander Willen
	
Date: 10/16/2020

*******************************************************************************/

/*******************************************************************************

I. Preliminaries

*******************************************************************************/

***A. Erase existing output if necessary
cap erase "$tables/Table_X4_Panel_A.xls"
cap erase "$tables/Table_X4_Panel_A.txt"
cap erase "$tables/Table_X4_Panel_B.xls"
cap erase "$tables/Table_X4_Panel_B.txt"


***B. Set control variables
local controls age age2 RACE2 RACE3 RACE4 RACE5

local fe i.day i.month i.year i.state

local ecp_controls hsdi scol cold young_children middle_children old_children part_age part_age2 part_hsdi part_scol part_cold part_RACE2 part_RACE3 part_RACE4 part_RACE5 part_immig part_secondgen part_yrsusa1 part_yrsusa2 part_cohort70_79 part_cohort80_89 part_cohort90_99 part_cohort00_09 part_cohort10_17

local wife_work wifehours wifehours_vary

/*******************************************************************************

II. Run Regressions

*******************************************************************************/

local maintit  "Table X4: Effect of parent source country characteristics for second-generation immigrants, prioritizing father source country characteristics"

**Loop through table panels
foreach panel in A B C D {
	
	*Set sample restrictions and labels
	preserve
	
	if "`panel'" == "A" {
		keep if female == 1 & secondgen==1
		
		local tabtit "`maintit', Panel A. All Women"
		
		}
		
	else if "`panel'" == "B" {
		keep if female == 0 & secondgen==1
		
		local tabtit "`maintit', Panel B. All Men"
		
		}
		
	else if "`panel'" == "C" {
		drop if hh_numownkids== 0
		keep if youngc<18 
		keep if female == 1 & secondgen==1
		
		local tabtit "`maintit', Panel C. Women with Children"
		
		}
		
	else if "`panel'" == "D" {
		drop if hh_numownkids== 0
		keep if youngc<18 
		keep if female == 0 & secondgen==1
		
		local tabtit "`maintit', Panel D. Men with Children"
		
		}


	**Loop through non-market work variables
	foreach var in $nmw $hwork $ccare { 
		
		if "`var'" == "$nmw" {
			local coltit "Total Non-Market Work"
			}
		else if "`var'" == "$hwork" {
			local coltit "Housework"
			}
		else if "`var'" == "$ccare" {
			local coltit "Childcare"
			}


		*Main specification from Tables 3 and 4 
		reg `var' parent_GGI_pf parent_GDP_000_pf parent_fertility_pf `controls' `ecp_controls' `fe' [aw=newwt], robust cluster(parent_bpl_pf)
		quietly summ `var' [aw=newwt] if e(sample)==1   
		outreg2 using "$tables/Table_X4_Panel_`panel'.xls", keep(parent_GGI_pf  parent_GDP_000_pf parent_fertility_pf ) append ctitle(`coltit') addstat(Mean, r(mean)) label title("`tabtit'") nocons
			
			
			}
	restore
	
	}
	
***Erase text files
cap erase "$tables/Table_X4_Panel_A.txt"
cap erase "$tables/Table_X4_Panel_B.txt"
cap erase "$tables/Table_X4_Panel_C.txt"
cap erase "$tables/Table_X4_Panel_D.txt"
